

## Notas técnicas – Tips de SAP Netweaver ABAP – Tip en breve

(Lo nuevo, lo escondido, o simplemente lo de siempre pero bien explicado)

"Tips en breve" se envía con frecuencia variable a la comunidad SAP. No contiene mensajes publicitarios de ninguna clase, sólo información técnica útil. Conteste el mensaje con REMOVE si no desea recibir más este servicio.

### VARIANTES VARIABLES

**Tema:** VARIANTE, FUNCIONES  
**Utilidad:** Configurar valores a los parámetros de un variante dentro de un programa ABAP.  
**Nivel:** Intermedio

Cuando se ejecuta, en forma interactiva, un programa ABAP, que posee parámetros, (parameters o select-options) es posible grabar las entradas ingresadas en dichos campos en un objeto llamado **Variante**. De esta forma, la próxima vez que se desee ejecutar el programa con los mismos valores se invoca a la variante generada.

En el caso de generar un programa que llama a otro en forma sometida (por ejemplo vía JOB\_SUBMIT) es necesario pasarle al programa llamado los valores de los parámetros deseados. Estos valores pueden guardarse, cada vez, en una variante e invocar al programa junto con el nombre de la variante que contiene los datos.

Se detalla a continuación, los elementos necesarios y la forma de utilización de la función **RS\_CHANGE\_CREATED\_VARIANT** que permite crear la variante y los valores de los parámetros del programa.

En el siguiente ejemplo, se completan los parámetros de Sociedad, Nro. de documento y Ejercicio en la variante de nombre ZBATCH del programa report ZJ\_1AF012.

Definir tablas internas donde se guardan los valores de los parámetros y son entrada a la función.

#### Ejemplo:

```
* tabla interna que contiene los valores de los parámetros
data: xvari like rsparams occurs 1 with header line .
data: xvarid like varid occurs 1 with header line .
```

Pasar los datos a los parámetros mediante la tabla interna. La estructura de la tabla interna es:

Nombre campo	Descripción breve
SELNAME	ABAP/4: Nombre de SELECT-OPTIONS / PARAMETERS
KIND	ABAP: Clase de selección
SIGN	ABAP: indicador para incluir/excluir valores
OPTION	ABAP: Opción de selección (EQ/B BT/CP/...)
LOW	ABAP: Valor de selección (LOW, HIGH, formato externo)

```
HIGH                ABAP: Valor de selección (LOW, HIGH, formato externo)
```

### Pasar los datos del documento a la variante

```
* SOCIEDAD
move: 'BR_BUKRS' to xvari-selname, "Nombre del param. en el pgm. llamado
      'S' to xvari-kind,          "Indicador de Select-option
      'I' to xvari-sign,          "Indicador de Inclusión
      'EQ' to xvari-option,       "Operador EQUAL
      zbkpf-bukrs to xvari-low.    "Valor del parámetro
append xvari.                    "Crea registro en tabla interna
* NRO. DOCUMENTO
move: 'BR_BELNR' to xvari-selname, "Nombre del param. en el pgm. llamado
      'S' to xvari-kind,          "Indicador de Select-option
      'I' to xvari-sign,          "Indicador de Inclusión
      'EQ' to xvari-option,       "Operador EQUAL
      zbkpf-belnr to xvari-low.    "Valor del parámetro
append xvari.                    "Crea registro en tabla interna
* EJERCICIO
move: 'BR_GJAHR' to xvari-selname, "Nombre del param. en el pgm. llamado
      'S' to xvari-kind,          "Indicador de Select-option
      'I' to xvari-sign,          "Indicador de Inclusión
      'EQ' to xvari-option,       "Operador EQUAL
      zbkpf-gjahr to xvari-low.    "Valor del parámetro
append xvari.                    "Crea registro en tabla interna
```

Nota: si el parámetro del programa llamado fuera creado por la sentencia parameters en el campo kind corresponde la letra 'P'. Se muestra un ejemplo con el campo de sociedad.

```
* SOCIEDAD
move: 'BR_BUKRS' to xvari-selname, "Nombre del param. en el pgm. llamado
      'P' to xvari-kind,          "Indicador de Parameter
      'I' to xvari-sign,          "Indicador de Inclusión
      'EQ' to xvari-option,       "Operador EQUAL
      zbkpf-bukrs to xvari-low.    "Valor del parámetro
append xvari.                    "Crea registro en tabla interna
```

### Llamada a la función.

```
* Carga nuevos valores a la VARIANTE DE ORDEN DE PAGO-----
call function 'RS_CHANGE_CREATED_VARIANT'
  exporting
    curr_report      = 'ZJ_1AF012' "Nombre del programa llamado
    curr_variant     = 'ZBATCH'    "Nombre de la variante
    vari_desc        = xvarid      "Nombre tabla interna
  tables
    vari_contents    = xvari
  exceptions
    illegal_report_or_variant = 1
    illegal_variantname      = 2
    not_authorized           = 3
    not_executed             = 4
    report_not_existent      = 5
    report_not_supplied      = 6
    variant_doesnt_exist     = 7
    variant_locked           = 8
    selections_no_match      = 9
    others                   = 10.
if sy-subrc ne 0.
  message id 'ZDESARROLLOS' type 'I' number 15 with zbkpf-belnr
    'Imposible crear VARIANTE'.
endif.
```